

NAG Fortran Library Routine Document

F07PSF (CHPTRS/ZHPTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07PSF (CHPTRS/ZHPTRS) solves a complex Hermitian indefinite system of linear equations with multiple right-hand sides, $AX = B$, where A has been factorized by F07PRF (CHPTRF/ZHPTRF), using packed storage.

2 Specification

```
SUBROUTINE F07PSF(UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)
ENTRY      chptrs (UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)
INTEGER    N, NRHS, IPIV(*), LDB, INFO
complex   AP(*), B(LDB,*)
CHARACTER*1 UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To solve a complex Hermitian indefinite system of linear equations $AX = B$, this routine must be preceded by a call to F07PRF (CHPTRF/ZHPTRF) which computes the Bunch–Kaufman factorization of A using packed storage.

If $UPLO = 'U'$, $A = PUDU^H P^T$, where P is a permutation matrix, U is an upper triangular matrix and D is an Hermitian block diagonal matrix with 1 by 1 and 2 by 2 blocks; the solution X is computed by solving $PUDY = B$ and then $U^H P^T X = Y$.

If $UPLO = 'L'$, $A = PLDL^H P^T$, where L is a lower triangular matrix; the solution X is computed by solving $PLDY = B$ and then $L^H P^T X = Y$.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates how A has been factorized as follows:

if $UPLO = 'U'$, $A = PUDU^H P^T$, where U is upper triangular;

if $UPLO = 'L'$, $A = PLDL^H P^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n , the order of the matrix A .

Constraint: $N \geq 0$.

- 3: NRHS – INTEGER *Input*
On entry: r , the number of right-hand sides.
Constraint: $\text{NRHS} \geq 0$.
- 4: AP(*) – **complex** array *Input*
Note: the dimension of the array AP must be at least $\max(1, N * (N + 1) / 2)$.
On entry: details of the factorization of A stored in packed form, as returned by F07PRF (CHPTRF/ZHPTRF).
- 5: IPIV(*) – INTEGER array *Input*
Note: the dimension of the array IPIV must be at least $\max(1, N)$.
On entry: details of the interchanges and the block structure of D , as returned by F07PRF (CHPTRF/ZHPTRF).
- 6: B(LDB,*) – **complex** array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.
On entry: the n by r right-hand side matrix B .
On exit: the n by r solution matrix X .
- 7: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F07PSF (CHPTRS/ZHPTRS) is called.
Constraint: $\text{LDB} \geq \max(1, N)$.
- 8: INFO – INTEGER *Output*
On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\text{INFO} < 0$

If $\text{INFO} = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b , the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$|E| \leq c(n)\epsilon P|U||D||U^H|P^T, \text{ if UPLO = 'U',}$$

$$|E| \leq c(n)\epsilon P|L||D||L^H|P^T, \text{ if UPLO = 'L',}$$

$c(n)$ is a modest linear function of n , and ϵ is the **machine precision**.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|x\|_\infty} \leq c(n) \text{cond}(A, x)\epsilon$$

where $\text{cond}(A, x) = \| |A^{-1}| |A| |x| \|_\infty / \|x\|_\infty \leq \text{cond}(A) = \| |A^{-1}| |A| \|_\infty \leq \kappa_\infty(A)$. Note that $\text{cond}(A, x)$ can be much smaller than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07PVF (CHPRFS/ZHPRFS), and an estimate for $\kappa_{\infty}(A)$ ($= \kappa_1(A)$) can be obtained by calling F07PUF (CHPCON/ZHPCON).

8 Further Comments

The total number of real floating-point operations is approximately $8n^2r$.

This routine may be followed by a call to F07PVF (CHPRFS/ZHPRFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07PEF (SSPTRS/DSPTRS).

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} -1.36 + 0.00i & 1.58 + 0.90i & 2.21 - 0.21i & 3.91 + 1.50i \\ 1.58 - 0.90i & -8.87 + 0.00i & -1.84 - 0.03i & -1.78 + 1.18i \\ 2.21 + 0.21i & -1.84 + 0.03i & -4.63 + 0.00i & 0.11 + 0.11i \\ 3.91 - 1.50i & -1.78 - 1.18i & 0.11 - 0.11i & -1.84 + 0.00i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 7.79 + 5.48i & -35.39 + 18.01i \\ -0.77 - 16.05i & 4.23 - 70.02i \\ -9.58 + 3.88i & -24.79 - 8.40i \\ 2.98 - 10.18i & 28.68 - 39.89i \end{pmatrix}.$$

Here A is Hermitian indefinite, stored in packed form, and must first be factorized by F07PRF (CHPTRF/ZHPTRF).

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07PSF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NRHMAX, LDB
      PARAMETER        (NMAX=8,NRHMAX=NMAX,LDB=NMAX)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N, NRHS
      CHARACTER        UPLO
*      .. Local Arrays ..
      complex          AP(NMAX*(NMAX+1)/2), B(LDB,NRHMAX)
      INTEGER          IPIV(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
      EXTERNAL         chptrf, chptrs, X04DBF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07PSF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX) THEN
*
*          Read A and B from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
          READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
          READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
```

```

      END IF
      READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*      Factorize A
*
      CALL chptrf(UPLO,N,AP,IPIV,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*          Compute solution
*
      CALL chptrs(UPLO,N,NRHS,AP,IPIV,B,LDB,INFO)
*
*          Print solution
*
      IFAIL = 0
      CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed','F7.4',
+              'Solution(s)','Integer',RLABS,'Integer',CLABS,
+              80,0,IFAIL)
      ELSE
      WRITE (NOUT,*) 'The factor D is singular'
      END IF
      END IF
      STOP
*
      END

```

9.2 Program Data

F07PSF Example Program Data

```

  4  2                                     :Values of N and NRHS
  'L'                                     :Value of UPLO
(-1.36, 0.00)
( 1.58,-0.90) (-8.87, 0.00)
( 2.21, 0.21) (-1.84, 0.03) (-4.63, 0.00)
( 3.91,-1.50) (-1.78,-1.18) ( 0.11,-0.11) (-1.84, 0.00) :End of matrix A
( 7.79,  5.48) (-35.39, 18.01)
(-0.77,-16.05) (  4.23,-70.02)
(-9.58,  3.88) (-24.79, -8.40)
( 2.98,-10.18) ( 28.68,-39.89)                :End of matrix B

```

9.3 Program Results

F07PSF Example Program Results

Solution(s)

```

           1           2
1  ( 1.0000,-1.0000) ( 3.0000,-4.0000)
2  (-1.0000, 2.0000) (-1.0000, 5.0000)
3  ( 3.0000,-2.0000) ( 7.0000,-2.0000)
4  ( 2.0000, 1.0000) (-8.0000, 6.0000)

```
